TEMPLATE = app

CONFIG(debug, debug|release) {
    TARGET = qedad
} else {
    TARGET = qeda
}

ROOTDIR = ../..
DESTDIR = $$ROOTDIR/bin

QT += \
    core \
    gui \
    xml

SOURCES += \
    appsettings.cpp \
    main.cpp \
    mainwindow.cpp \
    mainwindowslots.cpp \
    qtcreator/fancylineedit.cpp \
    qtcreator/filterlineedit.cpp \
    qlib/componentdialog.cpp \
    qlib/librarydock.cpp \
    qlib/librarytree.cpp \
    qlib/qedacomponent.cpp \
    qlib/qedalibrary.cpp \
    qlib/librarymodel.cpp \
    qlib/librarydialog.cpp \
    qlib/librarytreeitem.cpp \
    qlib/libraryfilter.cpp \
    qlib/componentwindow.cpp \
    qlib/componentgeneral.cpp \
    qlib/componentpinmanager.cpp

HEADERS += \
    appsettings.h \
    mainwindow.h \
    qedaversion.h \
    qtcreator/fancylineedit.h \
    qtcreator/filterlineedit.h \
    qlib/componentdialog.h \
    qlib/librarydock.h \
    qlib/librarytree.h \
    qlib/qedacomponent.h \
    qlib/qedalibrary.h \
    qlib/librarymodel.h \
    qlib/librarydialog.h \
    qlib/librarytreeitem.h \
    qlib/libraryfilter.h \
    qlib/componentwindow.h \
    qlib/componentgeneral.h \
    qlib/componentpinmanager.h

win32 {
RC_FILE += core.rc
QMAKE_LFLAGS += -static-libgcc
}
RESOURCES = core.qrc

TRANSLATIONS += \
    translations/core_ru.ts
CODECFORTR = UTF-8

# Pre build settings
# Create our custom translate target.
translate.commands = $$QMAKE_LIBDIR_QT/../bin/lrelease translations/core_ru.ts
QMAKE_EXTRA_TARGETS += translate
# Hook our translate target in between qmake's Makefile update and the actual project target.
translatehook.depends = translate
CONFIG(debug,debug|release):translatehook.target = Makefile.Debug
CONFIG(release,debug|release):translatehook.target = Makefile.Release
QMAKE_EXTRA_TARGETS += translatehook
